
ABSTRACT
    This paper displays a user example of an OS/390 Drill Down computer performance chart. The chart was output directly from a SAS/GRAPH® batch job to the OS/390 Webserver.  A test MXG® Computer Performance Database was used for input data.

    The Drill Down SAS/GRAPH® chart displayed was produced on an OS/390 system using SAS/GRAPH® Version 8.2 with the Output Delivery System – ODS and the SAS/Graph® GIF driver.  The WEB GIFS and HTML reports are updated daily on the mainframe intranet Webserver for viewing via a WEB browser. 

     OS/390® mainframe computer performance data is collected using IBM System Management Facilities (SMF) and Resource Measurement Facility (RMF) data.  The raw SMF/RMF data is processed with MXG® and SAS® software into an easily usable SAS® format.

INTRODUCTION 

    The SAS/MXG software processes the OS/390 SMF/RMF performance data into daily, weekly and monthly MXG® Computer Performance Databases.

    These databases contain numerous SAS® datasets.  Each SAS® dataset contains performance variables in a SAS® readable format.  The advantage of using MXG® software is that it converts raw performance data from IBM and third party computer vendors into SAS® readable format for processing by SAS® jobs.

   MXG® provides numerous examples of SAS® analyze examples and plots, from utilizing SAS/BASE to using SAS/GRAPH®.  


PREPARATION
    The Drill Down SAS/GRAPH® chart is output directly to a path to the OS/390 Webserver.  Of course you must first check with your Webserver technician(s) that the path exists and additionally that the path can also be accessed via your intranet web browser.  The usual path for initial testing is often /u/xxxxxxx where xxxxxxx is your userid.  The OS/390 TSO command panel can be used with the “ISH” or “OMVS” command to initially work on the Webserver path.  Setting up the Webserver to create or access the path isn’t part of this paper.  

     A macro was used to write the daily chart to a path with the day in the name. For this example, the path used is /sas/mxg/ddd.  Were ddd is the day sun, mon…sat.

   The SMF/RMF data was dumped at midnight by a SMFDUMP job submitted by the mainframe automated operations software.

Figure #1   OS/390  SAS  JCL 
//****************************************************

//*DOC:DAILY CHART DASD IO RATE                       

//****************************************************

//V8IORATE EXEC SASV8,

//  OPTIONS='SOURCE  NOMEMRPT MACRO',

//  WORK='400,200'

//INSTREAM DD UNIT=SYSDA,SPACE=(TRK,(1,1)),

//            DISP=(,PASS)

//LIBRARY  DD DSN=MXG.FORMATS,DISP=SHR

//SOURCLIB DD DSN=MXG.USER.SOURCLIB,DISP=SHR

//         DD DSN=MXG.SOURCLIB,DISP=SHR

//PDB      DD DSN=MXG.PDB.DAILY(0),DISP=SHR

//SYSIN DD DSN=MXG.USERGIF.SOURCLIB(V8IORATE),DISP=SHR

    MXG® also provides SAS/GRAPH® charts and plots with some of the MXG® DOCGRAF and GRAFXXXX examples as source code. 

Figure #2   OS/390 CHART DASD IORATE 

SAS/GRAPH® HOURLY DRILL DOWN VBAR3D GCHART OF DASD I/O RATE 
VBAR3D Drill Down Chart

    The SAS/GRAPH® VBAR3D chart displayed above (Figure #2) was created using the SAS Version 8 Output Delivery System ODS and PROC GCHART.  A test MXG® PDB was used for input data.

    The chart displays the OS/390 DASD I/O Rate for each hour of the day.  The OS/390 values for the computer system id and the date are taken from the input data and displayed in the chart title by using  #BYVALS.

    This drill-down chart is viewed with a web browser.  By using a mouse to click on a bar more detail can be viewed via the html drill down link for each hourly bar.

    The SAS® code to create the chart is displayed in Figure #4.  ODS produced the HTML body with the name of  “drill_down_dasdioratehour.html.”  This html file displays the chart as a GIF file and includes Java script to reference the drill-down html link for each bar.  

    The VBAR3D statement option HTML=variablename is used to create drill-down links.   In this example the code HTML=IODRILLDOWNLINKS is used.  The variable name  ‘IODRILLDOWNLINKS’ was assigned a different value for each hour in the DATA step.

    The GCHART program in this example creates links, but doesn’t create the actual drill-down file for each hour.
Figure #3   OS/390  PROC PRINT 

 PROC PRINT OUTPUT 

DRILL DOWN DETAILS - FOR HOUR = 15
Drill Down Details

        The DASD I/O counts for each DASD volser is in the MXG TYPE74 dataset.  This information is taken from the RMF type74 record on I/O activity.

    The above display and html file ‘dasdioprint15.html’ is the drill-down file for hour=15.  The previous GCHART program example created the VBAR3D chart, the BODY html with the drill-down links, but did not create the actual twenty-four hourly drill-down files.

    When you click on a bar on the chart, you will drill down to the respective html file if it exists.

    The SAS code used in Figure#6 created twenty-four hourly html drill-down files ‘dasdioprint00.html’ to ‘dasdioprint23.html’ for the daily chart.  This was accomplished by using a macro and ODS to create twenty-four (24) separate HTML files.

    This report included the DASD volser, I/O rate, I/O response time, and additional metrics in the type74 dataset including the duration or time interval.

    You have numerous options of what you want to include in each drill down report and the format of the report.

     The following SAS® code is used to create the VBAR3D Drill Down chart as a GIF named “v8iorate.gif.”

Figure #4  VBAR3D CHART DASD I/O RATE.

 /**  CHART DAILY DASD HOURLY IORATE         **/

 /**  JOB RUNS DAILY AND CREATES HTML AND GIF**/

 /** MEMBER=V8IORATE      IO HOUR DRILL DOWN **/

 OPTIONS  NODATE NONUMBER NOBYLINE LABEL ;

 %INCLUDE SOURCLIB(V8MACROS);

 %PICKPATH; /* PICK PATH THE PATH TO THE GIF  */

 FILENAME ODSOUT _DAY;  /* Path for ODS       */

DATA IOCOUNT;

FORMAT  IODRILLDOWNLINKS $VARYING100.;

  SET PDB.RMFINTRV;

      IF SYSTEM NE 'SYS7' THEN DELETE; /* SYS7*/

      IF DATE NE ZDATE-1  THEN DELETE;          

      IORATE=SIO74CNT/DURATM;

 /* CREATE A DRILL DOWN LINK FOR THE HOUR     */

 /* ie. HREF=/sas/mxg/mon/dasdioprint00.html  */

 /*                 to                          

 /* ie. HREF=/sas/mxg/mon/dasdioprint23.html */

 /* day part of link changes daily via  macro */

   IODRILLDOWNLINKS =  'HREF=' || _DAY || '/dasdioprint' || PUT(HOUR,Z2.0) || '.html';

  RUN;

 ODS LISTING CLOSE;

 ODS HTML PATH=ODSOUT  

   BODY='drill_down_dasdioratehour.html'

   (TITLE='SAS/MXG DASD I/O DRILL DOWN CHART');

 GOPTIONS RESET=GLOBAL DEVICE=GIF GUNIT=PCT

 XMAX=7.0IN YMAX=4.0IN  FTEXT=SWISSB

 CTEXT=MAGENTA  HTEXT=2.0  CBACK=CXFFFF00 ;
 PROC GCHART DATA=IOCOUNT; BY SYSTEM  DATE;

 FORMAT DATE WEEKDATE29.;

 FORMAT IORATE 5.0;

 LABEL  IORATE=’DASD*I/O*PER*SEC’;

   TITLE1 HEIGHT=3.9 C=BLACK

         'OS/390 COMPUTER SYSTEM = #BYVAL1';

   TITLE2 HEIGHT=3.8 C=BLACK 

         'DASD I/O ACTIVITY RATE PER SECOND';

   TITLE3 HEIGHT=3.8 C=RED  '#BYVAL2';

AXIS1 WIDTH=4  C=BLUE

  LABEL=(HEIGHT=3.0 C=VIB) /*HORIZ HOUR AXIS */

  VALUE=(HEIGHT=2.5 C=BROWN);

AXIS2 WIDTH=4  C=BLACK  /*RESPONSE IO AXIS */

       LABEL=(HEIGHT=2.4 COLOR=RED

       ROTATE=90 ANGLE=-90 )

       MAJOR=(HEIGHT=2 COLOR=GREEN)

       VALUE=(HEIGHT=2.9)

       ;


 FOOTNOTE1 HEIGHT=1.8 C=BLACK

        J=L 'SOURCE: MXG PERFORMANCE DATABASE'

        J=R C=BLACK 'SAS/GRAPH CHART';

 FOOTNOTE2 HEIGHT=1   J=R C=RED 'V8IORATE';

 PATTERN1 VALUE=SOLID COLOR=CXFF3300 ;

 NOTE H=2.6 PCT COLOR=GREEN BCOLOR=YELLOW

   FONT=SIMPLEX MOVE=(20,72) BOX=2 BSPACE=2

'DRILL DOWN CHART - CLICK ON A BAR FOR DETAILS';

 VBAR3D HOUR /  TYPE=MEAN SUMVAR=IORATE

   MEAN            /* display iorate on bar */

   DISCRETE        /* each bar displays     */

   SHAPE=PRISM     /* Shape of bars         */

   COUTLINE=BLACK  /* outline on bars       */

   CFRAME=CX33FFFF /* color inside frame    */

   HTML=IODRILLDOWNLINKS /*link for eachbar */

   NAME='v8iorate' /* name of gif           */

   MAXIS=AXIS1     /* hour axis             */

   RAXIS=AXIS2 ;   /* response axis         */

 RUN;


    The “PICKPATH” macro is one of several user macro’s in my sourclib member ‘‘V8MACROS.”  The macro is used in the previous code to assign a path for the charts and HTML.  

    The macro creates a  _DAY macro that appends the lowercase value of the day to the path.  

Figure #5    PICKPATH  SAS Macro
%MACRO PICKPATH ;                   

  /* USE DATA _NULL_ TO DEFINE _DAY MACRO */  

  DATA _NULL_;                                

  FORMAT WEBSERVER_PATH $CHAR50.;             

  TODAY=TODAY();                              

  YESTERDAY=TODAY-1;                          

  /* DAY TAKES YESTERDAY VALUE MON, TUE...*/  

  DAY= LOWCASE(PUT(YESTERDAY,WEEKDATE3.));    

  WEBSERVER_PATH = '/sas/mxg/' || DAY ; 

  /* put path in quotes & Delete blanks */    

  WEBSERVER_PATH = QUOTE(TRIM(WEBSERVER_PATH)); 

  /* _DAY MACRO POINTS TO PATH AND DAY  */ 

  FILE INSTREAM RECFM=FB LRECL=80 BLKSIZE=800;

  PUT @1  'MACRO _DAY '                        

      @12  WEBSERVER_PATH                      

      @63  '%' ;                               

  RUN;                                         

  %INCLUDE INSTREAM;                           

  RUN;                                         

  %MEND  PICKPATH;                             

Figure #6  ODS & PROC PRINT .

/*  MEMBER=V8DASDPR     PRINT DAILY DASD IO */

/** JOB RUNS DAILY AND CREATES HTML REPORTS**/

/** HTML REPORTS REFERENCED BY DRILL  CHART */

 OPTIONS NODATE NONUMBER  LABEL ;

 %INCLUDE SOURCLIB(V8MACROS);

 %PICKPATH;

 FILENAME ODSOUT _DAY;

 %LET SYSSEL='SYS7'; /* SYSTEM SELECTED */

 %MACRO PRINTHOURLY (HOURPRT, DRILLHTML);

 ODS LISTING CLOSE;

 ODS HTML PATH=ODSOUT STYLE=BROWN

          BODY=&DRILLHTML

          (TITLE='SAS/MXG I/O DETAILS');
 /* PRINT OUT ON TOP 5 DASD FOR THE HOUR */

 PROC PRINT SPLIT="*" 

   DATA=SORTOUT(OBS=5 WHERE=(HOUR=&HOURPRT) )

   STYLE(OBSHEADER)=

     {FOREGROUND=BLACK BACKGROUND=PINK}

   STYLE(HEADER)=

     {FOREGROUND=MAROON BACKGROUND=CXCCFFFF}

   STYLE(DATA)={FONT_SIZE=4 FONT_WEIGHT=BOLD

      FOREGROUND=BLUE BACKGROUND=YELLOW};

    BY SYSTEM DATE HOUR;

  FORMAT SIO74CNT COMMA10.0;

  FORMAT DURATM   TIME8.0;

  FORMAT TIME     TIME5.0;

  TITLE1 COLOR=BLUE BCOLOR=YELLOW BOX=4

 'OS/390 -TOP 5 DASD - DRILL DOWN DETAILS' ;

  VAR VOLSER TIME IORATE LCU AVGRSPMS DURATM  ;

 RUN;

 ODS HTML CLOSE;

 RUN;

 %MEND PRINTHOURLY;   /* END OF PRINT MACRO */

 /* Read TYPE74 I/O data for System selected*/

DATA IOCOUNT;

FORMAT DATE WEEKDATE29.;

  SET PDB.TYPE74;

      /* ONLY LOOK AT System Selected */

      IF SYSTEM NE &SYSSEL THEN DELETE;

      HOUR = HOUR(TIMEPART(STARTIME));

      DATE = DATEPART(STARTIME);

      TIME = TIMEPART(STARTIME);

      IF DATE NE ZDATE-1 THEN DELETE;

  RUN;

  PROC SORT DATA=IOCOUNT OUT=SORTOUT;

       BY SYSTEM DATE HOUR DESCENDING IORATE;

  RUN;

 /* PRINT OUT 24 FILES -HOUR and HTML FILE */

  %PRINTHOURLY(00,'dasdioprint00.html');

  %PRINTHOURLY(01,'dasdioprint01.html');

    “”    “”  

  %PRINTHOURLY(23,'dasdioprint23.html');

 RUN;



    The SAS code in Figure#6 used ODS and PROC PRINT to create the twenty-four (24) hourly drill-down HTML files named “dasdioprint00.html” to “dasdioprint23.html.”  Each HTML file is an hourly printed detail report.

    Again as in the previous code, the pickpath macro was utilized to define a _DAY macro.  This permits all of yesterday’s charts, html and gifs to be written to a day of the week path, i.e., /sas/mxg/sun, /sas/mxg/mon, etc.

    The program example also sorts the data by descending IORATE before printing.  This is done to have the print out display the DASD volser’s with the highest IORATES.

    The PROC PRINT uses SAS Version 8.2 and styles for bolder print and colored backgrounds in the header, observation header and for the data fields.  Although this is a basic use of styles, it makes for a more interesting display.

    The title also uses colors.  The text is a blue color on a yellow background.  Since by values are used the system id, date and hour are displayed on the byline.

    The daily MXG® pdb.type74 data was used for input to the program.  Detailed I/O activity on each DASD volser is recorded in the OS/390 RMF type74 record.  The dumped OS/390 SMF/RMF data is processed and stored in SAS® format in the daily MXG® PDB.TYPE74 dataset via the daily MXG PDB build process.


Note:  A modified test MXG/PDB was used for the I/O rates and DASD volser names displayed.  The computer system id, date, I/O rates and DASD volser names, etc, were changed from actual values for security reasons – not to include the actual values of detailed computer performance data in the paper.

CONCLUSION

    The updated daily SAS/GRAPH® Drill Down GIFS and Base SAS® HTML reports from the MXG®  computer performance database improved the ease of use for managers and technicians to review daily performance activity.

    The Drill Down feature also provides more interactive feedback with the data, which is appealing to many users and management.   The results can be easily interpreted by tactical performance technicians for investigating “bottlenecks” or by management to review for overall system performance and trends.

    Providing SAS/GRAPH® charts and BASE SAS® HTML reports from the SAS® Print procedures via the intranet also increased productivity of the staff by eliminating searches for lost or miss-routed printed reports.

